
AVERTISSEMENT

Ce document est la traduction francaise intégrale d’un document publié par Tomasz Grysztar sur le site flatas-
sembler.net. Malgré le soin apporté a cette démarche, elle n’est pas forcément exempte d’approximations ou
d’erreurs. Le lecteur est donc invité a se référer au texte original en cas d’incompréhension ou de doute.




1.  Qu'est-ce que I'assembleur flat G ?

Il s'agit d'un moteur d'assemblage congu pour succéder a celui utilisé dans I'assembleur flat 1, un assembleur
reconnu pour les processeurs x86. Ce moteur, de base, est incapable de reconnaitre et d'encoder les instructions
d'un processeur, mais il peut étre transformé en assembleur pour toute architecture de processeur. Son langage de
macro-instructions est nettement amélioré par rapport a celui de 'assembleur flat 1 et permet d'implémenter
facilement des encodeurs d'instructions sous forme de macro-instructions personnalisables.

Le code source de cet outil peut étre compilé avec 1'assembleur flat 1, mais il est également possible d'utiliser
l'assembleur flat G. Le code source contient des clauses incluant différents fichiers d'en-téte selon l'assembleur
utilisé. Lors de sa compilation, 1'assembleur flat G utilise 1'ensemble d'en-tétes fourni, qui implémente les ins-
tructions et formats x86 avec une syntaxe largement compatible avec I'assembleur flat 1.

Les exemples de programmes pour l'architecture x86 inclus dans ce pack sont des extraits initialement fournis
avec 1’assembleur flar 1. Ils utilisent des ensembles d'en-tétes implémentant les encodeurs d'instructions et les
formateurs de sortie nécessaires a leur assemblage, tout comme le faisait 1’assembleur flat d'origine.

Pour illustrer l'implémentation des jeux d'instructions de différentes architectures, des exemples de programmes
sont fournis pour les microcontréleurs 8051 et AVR. Leur simplicité les rend inadaptés a la programmation de
tels processeurs, mais ils constituent une base solide pour la création d'environnements de ce type.

Un exemple d'assemblage du code JVM est ¢galement fourni. Il s'agit d'une conversion de 1'exemple initialement
créé pour I’assembleur flat 1. De ce fait, cet exemple est quelque peu rudimentaire et n'exploite pas pleinement
les capacités du nouveau moteur. Il permet néanmoins de bien visualiser la structure d'un fichier de classe.

2.  Comment cela marche-t-il ?

La fonction essentielle de I'assembleur flat G est de générer une sortie définie par les instructions du code source.
Pour une ligne de texte comme celle présentée ci-dessous, I'assembleur générerait un seul octet avec la valeur
indiquée :

db 90h

Les macro-instructions peuvent étre définies pour générer des séquences de données spécifiques en fonction des
parametres fournis. Elles peuvent correspondre aux instructions du langage machine choisi, comme dans
I'exemple suivant, mais elles peuvent également étre définies pour générer d'autres types de données, a des fins
diverses.

macro int number
if number = 3

db 0CCh
else
db 0CDh, number
end if
end macro
int 20h ; génére 2 octets

L'assembleur, tel qu'on le congoit, peut étre considéré comme un langage interprété, et l'assembleur présente
effectivement de nombreuses caractéristiques d’un interpréteur. Cependant, il partage également certains aspects
avec un compilateur. Il est possible qu'une instruction utilise une valeur définie plus loin dans le code source et
qu'elle dépende des instructions qui la précédent, comme le montre I'exemple suivant :
macro jmpi target
if target-($+2) < 80h & target-($+2) >= -80h
db OEBh
db target-($+1)

else
db 0E9h
dw target-($+2)
end if
end macro
jmpi start

db ' some data’
start:

L'instruction “jmpi”, définie précédemment, génére le code d'une instruction de saut, conformément a l'architec-
ture 8086. Ce code contient le décalage relatif de la cible du saut, stocké sur un octet ou un mot de 16 bits. Le
décalage relatif est calculé comme la différence entre 1'adresse de la cible et I'adresse de l'instruction suivante. Le



symbole spécial “$” indique l'adresse de l'instruction courante et sert a calculer le décalage relatif et a déterminer
s'il peut tenir sur un seul octet.

Ainsi, le code généré par “jmpi start” dans l'exemple ci-dessus dépend de la valeur de I'adresse “start”, qui
dépend elle-méme de la longueur de la sortie de toutes les instructions précédentes, y compris le saut en ques-
tion. Ceci crée une boucle de dépendances, et 1'assembleur doit trouver une solution respectant toutes les con-
traintes imposées par le code source. Cela serait impossible si 1'assembleur était un simple interpréteur impératif.
Son langage est donc, a certains égards, déclaratif.

Trouver une solution a de telles dépendances circulaires peut ressembler a la résolution d'une équation, et il est
méme possible de construire un exemple ou I'assembleur flat G est effectivement capable d'en résoudre une :

X = (x-1)*(x4+2)/2-2%(x+1)
db x

La référence circulaire a été réduite ici a une simple définition qui se référence elle-méme pour construire la
valeur. L'assembleur flat G parvient & trouver une solution dans ce cas, bien qu'il puisse échouer dans de nom-
breux autres situations. La méthode employée par cet assembleur consiste a effectuer plusieurs passages sur le
texte source, puis a tenter de prédire toutes les valeurs a partir des connaissances ainsi acquises. Cette approche
est généralement suffisante pour 'assemblage de code machine, mais rarement pour résoudre des équations com-
plexes ; I'exemple ci-dessus constitue une exception.

3.  Moyens d'analyse des arguments d'une instruction

Toutes les instructions ne possédent pas une syntaxe aussi simple que celle des exemples précédents. Pour facili-
ter le traitement des arguments pouvant contenir des constructions particulieres, 1'assembleur flat G fournit plu-
sieurs outils performants, illustrés ci-dessous par des exemples implémentant quelques instructions du processeur
Z80. Les régles d'utilisation de ces fonctionnalités sont décrites dans le manuel.

Lorsqu'une instruction accepte un nombre trés restreint d'arguments, chacun d'eux peut étre traité individuelle-
ment grace a la construction “match” :

macro EX? first, second
match (=SP?), first
match =HL?, second
db 0E3h
else match =IX?, second
db 0DDh, 0E3h
else match =1Y?, second
db 0FDh, OE3h
else
err "incorrect second argument"
end match
else match =AF?, first
match =AF"?, second
db 08h
else
err "incorrect second argument"
end match
else match =DE?, first
match =HL?, second

db 0EBh
else
err "incorrect second argument"
end match
else
err "incorrect first argument"
end match
end macro
EX (SP), HL
EX (SP), IX
EX AF, AF
EX DE, HL

Le caractére "?" apparait a de nombreux endroits pour indiquer que les noms ne sont pas sensibles a la casse ;
toutes ces occurrences pourraient étre supprimées afin de simplifier davantage 1’exemple.



Lorsque I’ensemble des valeurs possibles d’un argument est plus vaste mais présente certaines régularités, des
substitutions textuelles peuvent étre définies pour remplacer certains symboles par des constructions soigneuse-
ment choisies, qui peuvent ensuite étre reconnues et analysées.

A? equ [:111b:

]
B? equ [:000b:]
C? equ [:001b:]
D? equ [:010b:]
E? equ [:011b:]
H? equ [:100b:]
L? equ [:101b:]

macro INC? argument

match [:r:], argument
db 100b + r shl 3

else match (=HL?), argument
db 34h

else match (=IX?+d), argument
db 0DDh, 34h, d

else match (=1Y?+d), argument
db 0FDh, 34h, d

else
err "incorrect argument"
end match
end macro
INC A
INC B
INC (HL)
INC (IX+2)

Cette approche présente un inconvénient : elle permet d’utiliser directement une expression comme "[:0:]" dans
un argument. 1 est toutefois possible d’empécher une telle exploitation de la syntaxe en utilisant un préfixe dans
la construction “match” :

REG. A? equ [:111Db:]
REG. B? equ [:000Db: ]
REG. C? equ [:001Db:]
REG. D? equ [:010Db: ]
REG.E? equ [:011Db:]
REG. H? equ [:100Db: ]
REG.L? equ [:101b:]

macro INC? argument

match [:r:], REG.argument
db 100b + r shl 3

else match (=HL?), argument
db 34h

else match (=IX?+d), argument
db 0DDh, 34h, d

else match (=I1Y?+d), argument
db 0FDh, 34h, d

else
err "incorrect argument"
end match
end macro

Dans le cas d'un argument structuré comme "(IX+d)", il peut étre souhaitable d'autoriser d'autres formes algébri-
quement équivalentes de l'expression, telles que "(d+I1X)" ou "(ct+IX+d)". Plutét que d'analyser individuellement
chaque variante possible, il est possible de laisser I'assembleur évaluer I'expression en traitant le symbole sélec-
tionné d'une manicre particuliere. Lorsqu'un symbole est déclaré comme un “élément”, il n'a aucune valeur ;
lorsqu'il est utilisé dans une expression, il est traité algébriquement comme un terme variable dans un polynéme.

element HL?
element IX?



element IY?

macro INC? argument
match [:r:], argument
db 100b + r shl 3
else match (a), argument
if a eq HL
db 34h
else if a relativeto [X
db 0DDh, 34h, a-1X
else if a relativeto IY
db 0FDh, 34h, a-1Y

else
err "incorrect argument"
end if
else
err "incorrect argument"
end match
end macro

INC (3%8+IX+1)

virtual at IX
x db ?
y db ?

end virtual

INC (y)

Il y a un petit probléme avec la macro-instruction ci-dessus. Un paramétre peut contenir n'importe quel texte et,
lorsqu'une telle valeur est insérée dans une expression, cela peut induire un comportement erratique. Par
exemple, si "INC (1/0)" était traité, I'expression "a eq HL" deviendrait "1|0 eq HL", et cette expression logique
est correcte et vraie méme si 'argument est mal formé. Cet effet secondaire indésirable est une conséquence du
fonctionnement des macro-instructions selon un principe simple de substitution de texte (et la meilleure fagon
d'éviter ce genre de problémes est d'utiliser CALM). Pour éviter cela, on peut utiliser une variable locale comme
proxy pour stocker la valeur d'un argument.

macro INC? argument
match [:r:], argument
db 100b + r shl 3
else match (a), argument
local value
value = a
if value eq HL
db 34h
else if value relativeto IX
db 0DDh, 34h, a-1X
else if value relativeto IY
db OFDh, 34h, a-1Y

else
err "incorrect argument"
end if
else
err "incorrect argument"
end match
end macro

L'utilisation d'une variable proxy présente un avantage supplémentaire : sa valeur est calculée avant que la ma-
cro-instruction ne génére une quelconque sortie. Lorsqu'une expression contient un symbole comme “$”, sa va-
leur peut varier selon l'endroit ou il est calculé. L'utilisation d'une variable proxy garantit que la valeur utilisée
est bien celle obtenue en évaluant l'argument avant la génération du code de l'instruction.

Lorsque I'ensemble des symboles autorisés dans les expressions est vaste, il est préférable d'utiliser une construc-
tion unique pour traiter toute une famille de symboles. Une déclaration “element” permet d'associer une valeur
supplémentaire a un symbole. Cette information peut ensuite étre récupérée grace a l'opérateur “metadata” appli-

qué a un polyndme lin€aire contenant ce symbole comme variable. L'exemple suivant est une variante de la ma-
cro-instruction précédente illustrant 'utilisation de cette fonctionnalité :

element register

element A? : register + 111b
element B? : register + 000b
element C? : register + 001b
element D? : register + 010b
element E? : register + 011b
element H? : register + 100b

+

element L? : register + 101b

element HL?
element IX?
element IY?

macro INC? argument
local value
match (a), argument
value = a
if value eq HL
db 34h
else if value relativeto IX
db 0DDh, 34h, a-1X
else if value relativeto IY
db 0FDh, 34h, a-1Y
else
err "incorrect argument"
end if
else match any more, argument
err "incorrect argument"

else
value = argument
if value eq value element 1 & value metadata 1 relativeto register
db 100b + (value metadata 1 - register) shl 3
else
err "incorrect argument"
end if
end match
end macro

La spécification “any more” permet de détecter tout argument contenant une expression complexe composée de
plusieurs éléments. Ceci empéche I'utilisation de syntaxes telles que “INC A+0” ou “INC A+B-A”. Cependant, pour
certains jeux d'instructions, l'inclusion de cette contrainte peut relever d'une préférence personnelle.

La condition “value eq value element 1” garantit que la valeur ne contient aucun terme autre que le nom d'un
registre. Méme lorsqu'un argument est contraint de ne contenir qu'un seul élément, il est possible qu'il ait une
valeur complexe, par exemple avec des définitions telles que “X = A + B” ou“Y = 2 * A”. Dans ce cas, “INC X”
et “INC Y” entraineraient tous deux le renvoi de la valeur “A” par l'opérateur “element 17, ce qui différe de la
valeur vérifiée.

Si une instruction prend un nombre variable d'arguments, une méthode simple pour identifier ses différentes
formes consiste a déclarer un argument avec le modificateur “&” afin de transmettre l'intégralité des arguments a
3 2

match”.

element CC

NZ? := CC + 000b
Z? :=CC + 001b
NC? := CC + 010b
C? :=CC + 011b
PO := CC + 100b
PE :=CC + 101b
P :=CC + 110b
M :=CC+ 1llb



macro CALL? arguments&
local cc,nn
match condition =, target, arguments
cc = condition - CC
nn = target
db 0C4h + cc shl 3

else
nn = arguments
db 0CDh
end match
dw nn
end macro
CALL 0

CALL NC, 2135h

Cette approche permet également de gérer des cas plus complexes, comme lorsque les arguments contiennent
des virgules ou sont délimités de différentes manicres.

Les macros CALM offrent un contrdle plus précis du processus d'analyse des arguments, et leur variante “match”
propose davantage d'options. Voici comment la macro ci-dessus pourrait étre réécrite :

calminstruction CALL? target&
local condition
match condition:name =, target, target,

jno unconditional
check defined condition & condition relativeto CC
jno error

emit 1, 0C4h + (condition - CC) shl 3
jump address

error:

err "unrecognized syntax"
unconditional:

emit 1, 0CDh
address:

emit 2, target
end calminstruction

Cette fois, aucun proxy n'est nécessaire, car le texte des arguments n'est évalué que lors du calcul des expressions
qui les contiennent. De plus, la fonction “match” peut étre utilisée pour vérifier la présence d'expressions mal
formées, tout comme elle vérifie la validité du nom du symbole de condition dans l'exemple.

4. Traitement des étiquettes

Une méthode standard pour définir une étiquette consiste a la faire suivre du symbole 2 points (ceci fait office de
saut de ligne et toute autre commande, y compris une autre étiquette, peut suivre sur la méme ligne). Une telle
étiquette définit simplement un symbole dont la valeur est égale a I'adresse courante, initialement nulle et incré-
mentée a chaque ajout d'octets a la sortie.

Dans certaines variantes du langage assembleur, il peut étre souhaitable d'autoriser une étiquette a précéder une
instruction sans ":" supplémentaire. Il est alors nécessaire de créer une macro-instruction étiquetée qui, apres
avoir défini une étiquette, transmet le traitement a la macro-instruction d'origine portant le méme nom.

struc INC? argument

INC argument
end struc

start INC A
INC B

Il faut procéder ainsi pour chaque instruction nécessitant ce type de syntaxe. Une simple boucle comme la sui-
vante suffirait :
iterate instruction, EX, INC, CALL
struc instruction? argument
. instruction argument



end struc
end iterate

Chaque instruction intégrée définissant des données posséde déja sa variante étiquetée.

En définissant une instruction étiquetée avec “?” a la place du nom, il est possible d'intercepter toute ligne com-
mengant par un identifiant qui n'est pas une instruction connue et qui est donc considéré comme une étiquette.
[T 3]

L'exemple suivant autoriserait une étiquette sans “:” a commencer n'importe quelle ligne du texte source (il gere
“="" et d'une valeur fonctionnent toujours) :

1381

également les cas particuliers : les étiquettes suivies de “:” ou de

struc ? tail&
match :, tail

else match : instruction, tail
. instruction

else match == value, tail
= value

else
: tail

end match

end struc

De toute évidence, il n'est plus nécessaire de définir des macro-instructions étiquetées spécifiques lorsqu'un effet
global de ce type est appliqué. Une variante doit étre choisie en fonction du type de syntaxe a autoriser.

L'interception des étiquettes définies avec ":" peut s'avérer utile lorsque la valeur de I'adresse courante nécessite
un traitement supplémentaire avant d'étre affectée a une étiquette, par exemple lorsqu'un processeur utilise des
adresses dont 1'unité est supérieure a un octet. La macro-instruction d'interception pourrait alors ressembler a
ceci :
struc ? tail&
match :, tail
label . at § shr 1
else match : instruction, tail
label . at § shr 1

instruction
else
tail
end match
end struc

La valeur de l'adresse actuelle utilisée pour définir les étiquettes peut étre modifiée avec “org”. Si les étiquettes
doivent étre différenciées de valeurs absolues, un symbole défini avec “element” peut étre utilisé pour former
une adresse :

element CODEBASE
org CODEBASE + 0

macro CALL? argument
local value
value = argument
if value relativeto CODEBASE
db 0CDh
dw value - CODEBASE
else
err "incorrect argument"
end if

end macro

Pour définir des étiquettes dans un espace d'adressage qui ne sera pas affiché en sortie, il convient de déclarer un
bloc “virtuel”. L'exemple suivant prépare les macro-instructions “DATA” et “CODE” pour alterner entre la généra-
tion d'instructions de programme et d'étiquettes de données. Seuls les codes d'instruction seront affichés.

element DATA

DATA_OFFSET = 2000h

element CODE

CODE_OFFSET = 1000h

macro DATA?



_END
virtual at DATA + DATA_OFFSET
end macro

macro CODE?

_END

org CODE + CODE_OFFSET
end macro

macro _END?
if $ relativeto DATA
DATA_OFFSET = $ - DATA
end virtual
else if $§ relativeto CODE
CODE_OFFSET = $ - CODE
end if
end macro

postpone
_END
end postpone

CODE

Le bloc “postpone” est utilisé ici pour garantir la fermeture correcte du bloc “virtual”, méme si le texte source
se termine par des définitions de données.

Dans I'environnement préparé par I'exemple ci-dessus, toute instruction peut distinguer les étiquettes de données
de celles définies dans le programme. Par exemple, une instruction de branchement peut accepter comme argu-
ment une étiquette du programme ou une valeur absolue, mais refuser toute étiquette de données.

macro CALL? argument
local value
value = argument
if value relativeto CODE
db 0CDh
dw value - CODE
else if value relativeto 0

db 0CDh
dw value
else
err "incorrect argument"
end if
end macro

DATA
variable db ?
CODE

routine:
Dans ce contexte, soit "CALL routine" soit "CALL 1000h" seraient autorisés, alors que "CALL variable" ne le serait
pas.
Lorsque les étiquettes ont des valeurs qui ne sont pas des nombres absolus, il est possible de générer des relocali-
sations pour les instructions qui les utilisent. Un bloc "virtuel" spécial peut étre utilisé pour stocker les déca-
lages des valeurs a l'intérieur du programme qui doivent étre déplacées lorsque sa base change :
virtual at 0
Relocations::
rw RELOCATION_COUNT
end virtual

RELOCATION_INDEX = 0



postpone
RELOCATION_COUNT := RELOCATION_INDEX
end postpone

macro WORD? value
if value relativeto CODE
store $§ - CODE : 2 at Relocations : RELOCATION_INDEX shl 1
RELOCATION_INDEX = RELOCATION_INDEX + 1
dw value - CODE
else
dw value
end if
end macro

macro CALL? argument
local value
value = argument
if value relativeto CODE | value relativeto 0

db 0CDh
word value
else
err "incorrect argument"
end if
end macro

Le tableau des relocalisations ainsi créé est ensuite accessible via la commande “load”. Les deux lignes sui-
vantes permettent d'afficher l'intégralité du tableau dans le résultat :

load RELOCATIONS : RELOCATION_COUNT shl 1 from Relocations : 0
dw RELOCATIONS

L'instruction “load” lit I'intégralité du tableau et le stocke dans une seule chaine de caracteres, puis l'instruction
“dw” I'écrit en sortie (avec un remplissage jusqu'a atteindre une longueur multiple d'un mot, mais dans ce cas
précis, la chaine ne nécessite jamais ce remplissage).

Pour des relocalisations plus complexes, des modificateurs supplémentaires peuvent étre nécessaires. Par
exemple, si les parties supérieure et inférieure d'une adresse doivent étre stockées séparément (probablement sur
deux instructions) et relocalisées séparément, les modificateurs nécessaires peuvent étre implémentés comme
suit :

element MOD. HIGH

element MOD. LOW

HIGH? equ MOD. HIGH +
LOW? equ MOD. LOW +

macro BYTE? value
if value relativeto MOD. HIGH + CODE
; register HIGH relocation
db (value - MOD. HIGH - CODE) shr 8
else if value relativeto MOD. LOW + CODE
; register LOW relocation
db (value - MOD.LOW - CODE) and OFFh
else if value relativeto MOD. HIGH
db (value - MOD. HIGH) shr 8
else if value relativeto MOD. LOW
db (value - MOD.LOW) and OFFh
else
db value
end if
end macro

Les commandes permettant d'enregistrer la relocalisation ont été omises par souci de clarté. Dans ce cas, il ne
s'agit pas seulement d'un décalage dans le code, mais aussi d'informations supplémentaires a enregistrer dans les



structures appropriées. Grace a cette préparation, des unités relocalisables dans le code pourraient étre générées
comme suit :

BYTE HIGH address
BYTE LOW address

Cette approche permet d'activer facilement la syntaxe avec des modificateurs dans toute instruction qui utilise en
interne la macro-instruction “byte” lors de la génération de code.

5.  Génération de plusieurs sections de fichier en paralléle

Ce moteur d'assemblage posséde une seule sortie principale qui doit étre générée séquenticllement. Cela peut
s'avérer problématique lorsque le fichier doit contenir des sections distinctes pour le code et les données, issues
de fragments entrelacés et potentiellement répartis sur plusieurs fichiers sources. Il existe cependant plusieurs
méthodes pour y remédier, toutes reposant, d'une maniére ou d'une autre, sur les capacités de référence anticipée
de l'assembleur.
Une approche naturelle consiste a définir le contenu de la section auxiliaire dans un bloc “virtual” et a le copier
a I'emplacement approprié dans la sortie en une seule opération. Une fois étiqueté, un bloc “virtual” peut étre
rouvert plusieurs fois pour y ajouter des données.

include ' 8086. inc

org 100h

jmp CodeSection

DataSection:

virtual
Data::
end virtual

postpone
virtual Data
load Data.OctetString : $ - $$ from $$
end virtual
end postpone

db Data. OctetString
CodeSection:
virtual Data

Hello db "Hello!", 24h
end virtual

mov ah, 9
mov dx, Hello
int 21h

virtual Data
ExitCode db 37h
end virtual

mov ah, 4Ch
mov al, [ExitCode]
int 21h

Cela conduit & une syntaxe relativement simple, méme sans macros supplémentaires.

Une autre méthode consiste a placer les ¢léments de la section dans des macros et a les exécuter a 1'emplacement
requis dans le code source. L'inconvénient de cette approche est que la détection des erreurs dans les définitions
peut s'avérer complexe.

Les techniques permettant d'ajouter facilement des éléments a une section générée en parallele peuvent égale-
ment étre trés utiles pour générer des structures de données telles que des tables de relocalisation. Au lieu des
commandes “store” utilisées précédemment lors de la démonstration du concept, des directives de données

classiques peuvent étre utilisées a l'intérieur d'un bloc “virtual” rouvert pour créer des enregistrements de relo-
calisation.

6.  Options pour ’analyse d'autres types de syntaxe

Dans certains cas, une commande que l'assembleur doit analyser peut commencer par un élément différent d'un
nom d'instruction ou d'une étiquette. Il peut s'agir d'un nom précédé d'un caractére spécial, comme . ” ou “!”, ou
d'une construction d'un type totalement différent. Il est alors nécessaire d'utiliser la macro “?” pour intercepter

des lignes entiéres de texte source et traiter toute syntaxe spéciale de ce type.

Par exemple, s'il était nécessaire d'autoriser une commande écrite sous la forme “. CODE”, il serait impossible de
lI'implémenter directement comme une macro-instruction, car le point initial fait que le symbole est interprété
comme un symbole local, et une instruction définie globalement ne pourrait jamais étre exécutée de cette ma-
niére. La macro-instruction d'interception apporte une solution :

macro ? line&
match .=CODE?, line
CODE
else match . =DATA?, line
DATA
else
line
end match
end macro

Les lignes contenant ““. CODE” ou . DATA” sont traitées ici de maniére a invoquer la macro-instruction globale
correspondante, tandis que toutes les autres lignes interceptées sont exécutées sans modification. Cette méthode
permet de filtrer toute syntaxe particuliére et de laisser 'assembleur traiter les instructions classiques.

Il arrive que la syntaxe non conventionnelle soit attendue uniquement dans une zone spécifique du texte source,
par exemple a l'intérieur d'un bloc aux limites définies. La macro-instruction d'analyse syntaxique doit alors étre
appliquée uniquement a cet endroit, puis supprimée par “purge” a la fin du bloc.
macro concise
macro ? line&
match =end =concise, line
purge ?
else match destt==src, line
ADD dest, src
else match dest-==src, line
SUB dest, src
else match dest==src, line

LD dest, src

else match destt++, line
INC dest

else match dest--, line
DEC dest

else match any, line
err "syntax error"
end match
end macro
end macro

concise
C=0
Bt++
At=2
end concise
Une macro-instruction définie de cette manicre n'intercepte pas les lignes contenant des directives contrdlant le
flux d'exécution, telles que “if” ou “repeat”, et ces directives restent utilisables librement a 1'intérieur d'un tel
bloc. Le comportement serait différent si la déclaration était de la forme “macro ?! 1ine&”. Dans ce cas, chaque
ligne serait interceptée sans exception.

Une autre option pour intercepter les commandes spéciales consiste a utiliser “struc ?” afin d'intercepter uni-
quement les lignes ne commengant pas par une instruction connue (le symbole initial étant alors considéré

comme une étiquette). Comme cette méthode ne teste que les commandes inconnues, elle devrait engendrer une
surcharge moindre sur 'assembleur.

struc (head) ? tail&
match .=CODE?, head

CODE tail
else
head tail
end match
end struc

Toutes ces approches dissimulent un piege subtil. Une étiquette définie avec : peut étre suivie d'une autre instruc-
tion sur la méme ligne. Si cette instruction (ici masquée dans le parameétre “tail”) est une directive de controle
comme “if”, la placer dans la clause “else” entrainera une imbrication incorrecte des blocs de contréle. Une
solution possible consiste a appeler le contenu de “tail” en dehors du bloc “match”. On pourrait par exemple
utiliser une macro spéciale :

struc (head) ? tail&
local invoker
match . =CODE?, head
macro invoker
CODE tail
end macro
else
macro invoker
head tail
end macro
end match
invoker
end struc

Une option plus simple consiste a appeler directement la ligne d'origine et, en cas de nécessité de la remplacer, a
la faire ignorer a l'aide d'un autre intercepteur de ligne (qui se libére immédiatement apres) :
struc (head) ? tail&
match . =CODE?, head

CODE tail
macro ? line&
purge ?
end macro
end match
head tail

end struc

Cependant, une bien meilleure fagon d'éviter ce genre d'écueils consiste a utiliser les instructions CALM plutdt
que les macros standard. I est ainsi possible de traiter les arguments et d'assembler la ligne originale ou modifiée
sans recourir a aucune directive de contréle. Les instructions CALM offrent également des performances nette-
ment supérieures, ce qui peut s'avérer particulieérement important dans le cas d'intercepteurs appelés quasiment a
chaque ligne du texte source.

7.  Controle du contexte de reconnaissance des identifiants de symboles

Lors de la conception d'une macro a usage général, il est important de s'assurer de son bon fonctionnement quel
que soit l'endroit ou elle est appelée. Bien que la directive “local” permette de créer des symboles privés pour
chaque instance de la macro sans interférer avec un environnement inconnu, il arrive qu'une variable doive étre
partagée entre plusieurs appels de la macro. Un symbole défini globalement convient généralement a cet effet,
mais il peut étre sujet a des interférences. Prenons l'exemple suivant :

GLOBAL_STATE = 0

macro state
db GLOBAL_STATE
end macro

macro switch value

GLOBAL_STATE = GLOBAL_STATE xor (value)
end macro

Cette solution présente quelques problémes. Premiérement, si le symbole nommé “GLOBAL_STATE” était égale-
ment utilisé ailleurs a d'autres fins, cela interférerait avec ces macros. Choisir un nom rare et spécifique pour une
variable globale peut atténuer le probléme (et s'il est nécessaire d'utiliser plusieurs symboles globaux, il est pré-
férable de créer un espace de noms dédié, de préférence avec un nom inhabituel), mais il serait préférable de
séparer ces symboles d'une manicre ou d'une autre.

Un autre probléme se manifesterait si les macros ci-dessus étaient utilisées a l'intérieur d'un bloc “namespace”,
car la définition de “GLOBAL_STATE” créerait alors un symbole dans l'autre espace de noms. Il existe une solution
simple a ce second probléme : il suffit d'ajouter un point a la fin de l'identificateur pour garantir que le symbole
défini est bien celui dont la valeur est accédée a droite :

GLOBAL_STATE. = GLOBAL_STATE xor (value)

11 existe cependant une autre option. Les arguments fournis a la macro conservent le contexte de tous les identifi-
cateurs présents dans leur texte (sauf indication contraire, lorsque le nom d'un argument est précédé de “&”).
L'exemple suivant illustre cet effet :

macro tester arg
namespace X

a=»0
db a
db arg
end namespace
end macro
a=3
tester a

Les deux définitions de données assemblées contiennent le méme texte : “db a”, mais la seconde interpréte “a”
dans le contexte de I’appel de la macro et génére 1’octet 3 au lieu de 0. Ceci est rendu possible grace a un méca-
nisme comparable & une coloration syntaxique. La valeur du paramétre “arg” posséde une propriété supplémen-
taire, telle qu’une couleur de texte, qui différencie la premiére occurrence de “db a” de la seconde. Ainsi, si I’on
définit une macro a I’intérieur d’une autre, le texte de cette derniére conservera sa coloration. L’exemple suivant
utilise ce mécanisme pour améliorer la définition des macros “state” / “switch” de ’exemple précédent :

macro setup variable
variable = 0

macro state
db variable
end macro

macro switch value
variable = variable xor (value)
end macro

end macro

setup GLOBAL_STATE

namespace Program
switch 8
GLOBAL_STATE = 0
State

end namespace

Lorsque le fichier “setup GLOBAL_STATE” est assemblé, la macro “switch” est définie avec un corps contenant le
texte :

GLOBAL_STATE = GLOBAL_STATE xor (value)
ou les deux instances de "GLOBAL_STATE" sont associées au contexte global. D’autres parties du texte ne sont
associées a aucune information supplémentaire de ce type. Cependant, lorsque le “switch 8” est assemblé, la
ligne générée par la macro est :

GLOBAL_STATE = GLOBAL_STATE xor (8)
et, cette fois, un autre texte coloré apparait : la valeur “8” indique le contexte d’appel de la macro “switch”.

La variante de “switch” et “state” ci-dessus utilise toujours la variable globale, quel que soit I’endroit ou elle est
appelée. Ceci est similaire au concept de fermeture présent dans de nombreux langages de programmation.
La variable reste néanmoins globale et d’autres portions de code peuvent toujours y accéder. Pour la rendre tota-
lement inaccessible, on peut utiliser la directive “local”, qui crée un paramétre spécial remplagant son nom par
le méme texte, mais coloré spécifiquement pour indiquer le contexte propre a I’instance de la macro.

macro setup

local variable
variable = 0

macro state
db variable
end macro

macro switch value
variable = variable xor (value)
end macro

end macro

setup

Ce type de transfert de contexte se produit également lors de la définition d'une variable symbolique. Lorsqu'un
symbole est défini avec “equ” ou “define”, sa valeur entiére est marquée par le contexte présent au moment de
sa définition. Lorsque le texte de cette valeur est affecté a un paramétre avec “match” ou “irpv”, les informations
contextuelles véhiculées par les fragments de texte sont préservées.

First:

define LIST .x

Second:
X =2

LIST equ LIST, .x

match values, LIST
display ‘values
db values

end match

La directive “display” indique que le texte extrait de la variable “LIST” avec la fonction “match” est . x, .x”.
Or, les textes de ces deux identifiants ont des contextes différents, et il s'avére que la liste contient deux valeurs
distinctes.

Lorsqu'une fonction “match” découpe le texte d'un identifiant en plusieurs parties, chacune conserve les informa-
tions contextuelles du texte original. Ainsi, si un identifiant est reconstitué a partir de fragments de texte prove-
nant de différentes sources, seul le contexte associé a la partie initiale influe sur la reconnaissance du symbole.
Cela permet de forcer la reconnaissance de n'importe quel symbole dans son contexte actuel en le faisant précé-
der d'un diése (#).
macro tester name
namespace my

name db ? ; symbole défini dans son espace de nom d’ origine
#name db ? ; symbole défini dans 1’ espace de nom "my"
end namespace

end macro

Si cette astuce ne suffit pas et qu'il est nécessaire de dépouiller un texte de toute information contextuelle, la
directive “rawmatch” permet précisément cela. Un effet similaire peut étre obtenu en convertissant la valeur d'un
parameétre en chaine de caracteres a l'aide de 1'opérateur puis en interprétant ce texte brut avec “eval”.

14



Un argument de macro préfixé par “&” n'ajoute aucune information contextuelle a sa valeur, mais ne supprime
pas non plus les informations déja présentes dans le texte ou certaines de ses parties. Cela permet de transmettre
du texte en toute sécurité sans perte d'information, mais aussi sans le contaminer avec un contexte indésirable,
comme le montre cet exemple en plusieurs étapes :

calminstruction (var) transparent_equ &val&
publish :var, val
end calminstruction

namespace Windows
EOL := string 0x0AOD
link equ EOL

end namespace

EOL := 0x0A

match WinEOL, Windows. link

list equ WinEOL, EOL ; contexte ajouté
list transparent_equ WinEOL, EOL ; pas de contexte ajouté
end match

namespace (64

EOL := 0x0D

irpv items, list
db items

end irpv

end namespace

Tous les éléments transmis a “db” contiennent le texte “EOL”, mais interprétés différemment : dans le contexte de
l'espace de noms “Windows”, d'un espace de noms global, et enfin comme un texte brut sans contexte (auquel cas
il finit par pointer vers la définition la plus proche, celle de 1'espace de noms “C64”).

8.  Définition d’une instruction partageant un nom avec I'une des directives prin-
cipales

Il peut arriver qu'un langage soit généralement facile a implémenter avec des macros, mais qu'il doive inclure
une commande portant le méme nom qu'une directive de 1'assembleur. Bien qu'il soit possible de redéfinir n'im-
porte quelle instruction par une macro, les macros elles-mémes peuvent nécessiter un acces a la directive d'ori-
gine. Pour permettre a une instruction portant le méme nom d'appeler une instruction différente selon le contexte,
le langage implémenté peut étre interprété dans un espace de noms contenant la macro de redéfinition, tandis que
toutes les macros nécessitant un accés a la directive d'origine devraient temporairement basculer vers un autre
espace de noms ou celle-ci n'a pas été redéfinie. Cela impliquerait que chaque macro de ce type encapsule son
contenu dans un bloc “namespace”.

11 existe cependant une autre astuce, liée a la maniére dont les textes des parameétres de macro ou des variables
symboliques préservent le contexte d'interprétation des symboles qu'ils contiennent (y compris 'espace de noms
de base et 1'étiquette parente pour les symboles commengant par un point).

Contrairement aux deux cas mentionnés, le texte d'une macro ne contient généralement pas d'informations sup-
plémentaires. Toutefois, si une macro est construite de maniere a inclure du texte ayant appartenu a un parametre
d'une autre macro ou a une variable symbolique, ce texte conserve les informations de contexte méme lorsqu'il
est intégré a une nouvelle macro. Par exemple :
macro definitions end?
namespace embedded
struc LABEL? size
match , size

else
label . : size
end match
end struc

macro EZND? name
end namespace
match any, name
ENTRYPOINT := name

end match
macro ?! line&
end macro
end macro
end macro

definitions end

start LABEL
END start

Le paramétre passé a la macro “definitions” peut sembler inopérant, puisqu'il remplace chaque occurrence de
“end” par le méme mot. Cependant, le texte issu de ce parametre est enrichi d'informations contextuelles sup-
plémentaires, et cet attribut est conservé lorsqu'il est intégré a une nouvelle macro. Grace a cela, la macro “LA-
BEL” peut étre utilisée dans un espace de noms ou l'instruction “end” a une signification différente, tandis que ses
occurrences font toujours référence au symbole de 1'espace de noms externe.

Dans cet exemple, le paramétre est insensible a la casse ; il remplacerait donc méme le “END” de l'instruction
“macro” censée définir un symbole dans l'espace de noms “embedded”. C'est pourquoi 1'identificateur a été scindé
a l'aide d'un opérateur de concaténation afin d'éviter qu'il ne soit reconnu comme un parametre. Cela ne serait pas
nécessaire si le paramétre était sensible a la casse (comme c'est généralement le cas).

On peut obtenir le méme résultat en utilisant des variables symboliques a la place des paramétres de macro,
grace a la fonction “match” qui permet d'extraire le texte d'une variable symbolique.

define link end
match end, link
namespace embedded
struc LABEL? size
match , size

else
label . : size
end match
end struc

macro END? name
end namespace
match any, name

ENTRYPOINT := name

end match
macro ?! line&
end macro

end macro

end match

start LABEL
END start

Cela ne fonctionnerait pas sans passer par une variable symbolique, car les parametres définis par des directives
de contréle comme “match” n'ajoutent pas d'informations contextuelles au texte, sauf si elles y sont déja pré-
sentes.

Les instructions CALM offrent une autre approche a ce type de problémes. Si un jeu d'instructions personnalisé
est entierement défini sous forme d'instructions CALM, il peut méme se passer d'acces aux directives de controle
d'origine. Toutefois, si une instruction CALM doit assembler une directive potentiellement inaccessible, la va-
riable symbolique passée a “assemble” doit étre définie avec le contexte approprié pour le symbole d'instruction.

9. Conversion d’une macro-instruction en CALM

Une macro-instruction classique se compose de lignes de texte prétraitées (en remplagant les noms des para-
métres par leurs valeurs correspondantes) a chaque appel de l'instruction, puis transmises a 1'assembleur. Par
exemple, la macro-instruction qui suit génére une seule ligne a assembler, en remplagant “number” par le texte
fourni par son unique argument :

macro octet value¥
db value
end macro

Une instruction CALM peut étre vue comme un préprocesseur personnalisé, écrit dans un langage spécifique.
Elle utilise diverses commandes pour traiter les arguments et générer les lignes a assembler. De maniére géné-
rale, elle peut simuler le fonctionnement d'un préprocesseur standard grace a la commande “arrange”. Apreés le
prétraitement de la ligne, elle doit également la transmettre explicitement a l'assembleur avec la commande “as-
semble”.

calminstruction octet value*
arrange value, =db value
assemble value

end calminstruction

Cela donne le méme résultat que la macro-instruction d'origine, car le prétraitement est identique. Cependant,
contrairement au texte de la macro-instruction, un motif fourni a “arrange” doit indiquer explicitement quels
jetons de nom doivent étre remplacés par leurs valeurs et lesquels (précédés de “="") doivent rester inchanggés.
Les jetons copiés du motif sont dépouillés de toute information contextuelle, tout comme le texte de la macro-
instruction n'en contient généralement aucune (tandis que les valeurs issues des arguments conservent le contexte
de reconnaissance dans lequel l'instruction a été lancée).

Il s'agit de la méthode de conversion la plus simple ; une simple séquence de commandes “arrange” et “as-
semble” permet de générer les mémes lignes que la macro-instruction d'origine. Il existe toutefois une exception :
lorsqu'une commande “locale” est exécutée par une macro-instruction, elle crée un paramétre prétraité avec une
valeur spéciale pointant vers un symbole de I'espace de noms propre a l'instance donnée de l'instruction.

macro pointer
local next
dd next
next:
end macro

Dans le cas de CALM, aucun espace de noms n'est disponible ; l'espace de noms local d'une instruction CALM
est partagé entre toutes ses instances. Par conséquent, si un nouveau symbole unique est nécessaire a chaque
appel de l'instruction, il doit étre construit manuellement. Une méthode simple consiste a ajouter un nombre
unique au nom.

global _uid = 0

calminstruction pointer
compute global_uid, global_uid + 1
local command
arrange command, =dd =next#global_uid
assemble command
arrange command, =next#global_uid:
assemble command

end calminstruction

Ici, la fonction “arrange” regoit une variable numérique et doit la remplacer par du texte. Cela fonctionne uni-
quement si la valeur est un nombre entier positif ou nul ; dans ce cas, “arrange” la convertit en un jeton texte
contenant la représentation décimale de ce nombre. Les lignes transmises a 1'assembleur contiendront donc des
identificateurs tels que “next#1”.

Alors que l'incrémentation du compteur global pourrait étre réalisée en préparant une commande assembleur
standard comme “global_uid = global_uid + 1” avec “arrange” et en la transmettant a 'assembleur, la com-
mande “compute” permet de le faire directement dans le processeur CALM. De plus, elle n'est alors pas affectée
par quoi que ce soit qui modifie le contexte de I'assembleur. Si I'instruction était définie comme inconditionnelle
et utilisée dans un bloc IF ignoré, “compute” exécuterait tout de méme sa tache, car 1'exécution des commandes
CALM est, tout comme le prétraitement standard, indépendante du flux principal de I'assembleur. De plus, les
références a “global_uid” pointent toujours vers le méme symbole : celui qui était dans la portée lors de la défi-
nition et de la compilation de I’instruction CALM. Par conséquent, 1’incrémentation de sa valeur avec “compute”
est plus fiable et prévisible.

De maniere similaire, I'ensemble des lignes définissant le label peut étre remplacé par une commande "publish".
Ici, la valeur de 1'étiquette (qui doit étre égale a 1'adresse apres I'assemblage de la ligne contenant "dd") doit étre
calculée en premier, car "publish" effectue uniquement l'affectation d'une valeur au symbole :

global_uid = 0
calminstruction pointer

compute global_uid, global_uid + 1
local symbol, command

arrange symbol, =next#global_uid
arrange command, =dd symbol
assemble command
local address
compute address, $
publish symbol:, address

end calminstruction

L'instruction CALM étant conditionnelle, la fonction “publish” qu'elle contient 1'est également. Elle fonctionne
donc correctement en remplacement de 1'assemblage d'une ligne avec une étiquette.

Bien qu'un compteur global présente plusieurs avantages, il peut étre modifié ; 1'utilisation d'un compteur local
peut donc parfois s'avérer préférable. Cependant, 1'espace de noms local de l'instruction CALM n'est générale-
ment pas accessible de l'extérieur, ce qui complique 1'initialisation de ce compteur. Une solution consiste a véri-
fier si le compteur a déja été initialisé a 1'aide de la commande “take”.

calminstruction pointer
local id
take id, id
jyes increment
compute id, 0
increment:
compute id, id + 1
local symbol, command
arrange symbol, =next#id
arrange command, =dd symbol
assemble command
local address
compute address, $
publish symbol:, address
end calminstruction

Mais cela ajoute des commandes qui sont exécutées a chaque appel de l'instruction. Une meilleure solution con-
siste a exploiter la possibilité de définir des instructions personnalisées traitées lors de la définition de l'instruc-
tion CALM :

calminstruction calminstruction?. init? var¥, val:0
compute val, val
publish var, val

end calminstruction

calminstruction pointer
local id
init id, 0
compute id, id + 1
local symbol, command
arrange symbol, =next#id
arrange command, =dd symbol
assemble command
local address
compute address, $
publish symbol:, address
end calminstruction
L'instruction personnalisée “init” est appelée lors de la définition de 1'instruction CALM (elle ne génére aucune
commande a exécuter par l'instruction définie ; elle devrait elle-méme utiliser des commandes “assemble” pour
générer des instructions a compiler). Elle regoit le nom d'une variable de la portée locale de l'instruction CALM
et utilise “publish” pour lui assigner une valeur numérique initiale.
Pour initialiser une variable locale avec une valeur symbolique, une instruction personnalisée encore plus simple
suffirait :
calminstruction calminstruction?. initsym? var¥, val&
publish var, val
end calminstruction
Le texte de I'argument “val” contient le contexte de reconnaissance de la définition de l'instruction CALM qui
contient l'instruction “initsym”, permettant ainsi de préparer un texte pour “assembler” contenant des références
a des symboles locaux :

18



calminstruction bigendian32? value
local command
initsym command, dd value
compute value, value bswap 4
assemble command

end calminstruction

Une fois compilée, cette instruction ne contient plus que deux commandes : “compute” et “assemble”. La valeur
du symbole local “command” est un texte interprété dans le méme contexte local et faisant référence au méme
symbole “value” que “compute”.

Cet exemple illustre un autre avantage de CALM par rapport aux macro-instructions standard : sa sémantique
stricte empéche divers comportements indésirables permis par une simple substitution de texte. Le texte de “va-
lue” sera évalué par “compute” comme une sous-expression numérique, signalant une erreur en cas de syntaxe
inattendue. Il est donc préférable de traiter les arguments entiérement via des commandes CALM et de n'utiliser
“assemble” que pour les instructions simples finales. Ces dernicéres peuvent méme é&tre éliminées lorsqu'une
commande CALM existe pour effectuer la méme tache.

calminstruction bigendian32? value
emit 4, value bswap 4
end calminstruction

Cela effectue des opérations complétement indépendamment du processus d'assemblage standard, donc si I'ins-
truction contenant était définie comme inconditionnelle, la sortie serait générée méme si elle était invoquée a
l'intérieur d'un bloc "if" ignoré ou lors de la définition d'une macro-instruction (et cela se produirait sans ajouter
de ligne a la définition).

